
6 Steps for Two-Level SEM 
 
Step 0:  Do a Type=Basic run to get the cluster information, intraclass correlations, and 
pooled-within covariance matrix and estimated between correlation matrix.   
 

Input:  basic, basicB. 
 
Step 1:  Do an approximate exploration of the model using the regular (total) sample 
covariance matrix.   
  

A. Try the hypothesized model ignoring clustering.   
Input: step1 

B. Check the change in SEs and chi-square by using Type = Complex.   
Input: step1Complex 

C. Explore the factor structure from scratch using EFA ignoring clustering. 
Input: step1EFA 

 
Step 2:  Do the modeling based on the pooled-within covariance matrix to get an 
approximation of within-level parameter estimates (here applied to the y’s) 
 
 Input: step2 
 
Step 3:  Investigate the size of the intraclass correlations and cluster sizes to see if 
proceeding to multilevel modeling is warranted.  See output from Step0. 
 
Step 4: Explore the between structure using the estimated between matrix from Step0.  
Do an EFA on the between correlation matrix.   
 

Input: step4 
 
Step 5:  Do the full two-level analysis with within and between structures based on steps 
2 and 4. This is the model in the User’s Guide example 9.8. 
 
 Input: step5 

EXAMPLE 9.8: TWO-LEVEL CFA WITH CONTINUOUS 
FACTOR INDICATORS WITH TWO BETWEEN FACTORS 
AND ONE WITHIN FACTOR 
 
 

TITLE: this is an example of a two-level CFA with 
continuous factor indicators with two 
factors on the within level and one factor 
on the between level 

DATA:    FILE IS ex9.8.dat; 
VARIABLE: NAMES ARE y1-y6 x1 x2 w clus; 
 WITHIN = x1 x2; 
          BETWEEN = w; 
           CLUSTER IS clus; 
ANALYSIS: TYPE IS TWOLEVEL; 
 ESTIMATOR = MLR; 
MODEL: 
 %WITHIN% 
 fw1 BY y1-y3; 
 fw2 BY y4-y6; 
 fw1 ON x1 x2; 
 fw2 ON x1 x2; 
 %BETWEEN% 
 fb BY y1-y6; 
 fb ON w; 

Basic.inp 
 
TITLE: this is an example of a two-level CFA with 
 continuous factor indicators with two 
 factors on the within level and one factor 
 on the between level 
 
    Basic run: estimate within and between matrices, 
               estimate intra class correlations 
 
DATA:    FILE IS ex9.8.dat; 
VARIABLE: NAMES ARE y1-y6 x1 x2 w clus; 
    usev = y1-y6; 
    CLUSTER IS clus; 
ANALYSIS: TYPE IS TWOLEVEL basic; 
 
Savedata: 
    sample = spw.dat; 
    sigb = estsigb.dat; 
 
 
 
basicb.inp 
 
TITLE: this is an example of a two-level CFA with 
 continuous factor indicators with two 
 factors on the within level and one factor 
 on the between level 
 
    Basic run: estimate within and between matrices, 
               estimate intra class correlations 
 
DATA:    FILE IS ex9.8.dat; 
VARIABLE: NAMES ARE y1-y6 x1 x2 w clus; 
    usev = y1-y6; 
    CLUSTER IS clus; 
ANALYSIS: TYPE IS TWOLEVEL basic; 
 
Savedata: 
 !   sample = spw.dat; 
    sigb = estsigb.dat; 
    type = corr; 



 
step1.inp 
 
TITLE: this is an example of a two-level CFA with 
 continuous factor indicators with two 
 factors on the within level and one factor 
 on the between level 
 
    Step 1A: Ignoring clustering 
 
DATA:    FILE IS ex9.8.dat; 
VARIABLE: NAMES ARE y1-y6 x1 x2 w clus; 
 Usev = y1-w;   
MODEL: 
 f1 BY y1-y3; 
 f2 BY y4-y6; 
 f1 ON x1 x2 w; 
 f2 ON x1 x2 w; 
 
Output: 
    standardized sampstat; 
 
 
step1complex.inp 
 
TITLE: this is an example of a two-level CFA with 
 continuous factor indicators with two 
 factors on the within level and one factor 
 on the between level 
 
    Step 1B: Ignoring clustering in the modeling, 
            but correcting SEs 
 
DATA:    FILE IS ex9.8.dat; 
VARIABLE: NAMES ARE y1-y6 x1 x2 w clus; 
 Usev = y1-w; 
    cluster = clus; 
 
Analysis: 
    type = complex; 
          
MODEL: 
 f1 BY y1-y3; 
 f2 BY y4-y6; 
 f1 ON x1 x2 w; 
 f2 ON x1 x2 w; 
 
Output: 
    standardized sampstat; 



 
 
step1efa.inp 
 
TITLE: this is an example of a two-level CFA with 
 continuous factor indicators with two 
 factors on the within level and one factor 
 on the between level 
 
    Step 1C: EFA ignoring clustering 
 
DATA:    FILE IS ex9.8.dat; 
VARIABLE: NAMES ARE y1-y6 x1 x2 w clus; 
 Usev = y1-y6; 
          
Analysis: 
    type = efa 1 3; 
    estimator = ml; 
 
 
 
step2.inp 
 
TITLE: this is an example of a two-level CFA with 
 continuous factor indicators with two 
 factors on the within level and one factor 
 on the between level 
 
    Step 2: CFA using the pooled-within matrix 
 
DATA:    FILE = spw.dat; 
    type = cova; 
    nobs = 890; ! n = N - C 
VARIABLE: NAMES ARE y1-y6; 
  
Model: 
    f1 by y1-y3; 
    f2 by y4-y6; 
 
Output: 
   sampstat standardized; 



 
 
 
step4.inp 
 
 TITLE: this is an example of a two-level CFA with 
 continuous factor indicators with two 
 factors on the within level and one factor 
 on the between level 
 
    Step 4: EFA using the estimated between  
            correlation matrix 
 
DATA:    FILE = estsigb.dat; 
    type = corr; 
    nobs = 110; 
VARIABLE: NAMES ARE y1-y6; 
 
Analysis: 
    type = efa 1 3; 
    estimator = ml; 
 
Output: 
    sampstat; 
 
 
step5.inp 
 
TITLE: this is an example of a two-level CFA with 
 continuous factor indicators with two 
 factors on the within level and one factor 
 on the between level 
 
    Step 5: full 2-level analysis 
 
DATA:    FILE IS ex9.8.dat; 
VARIABLE: NAMES ARE y1-y6 x1 x2 w clus; 
 WITHIN = x1 x2; 
          BETWEEN = w; 
           CLUSTER IS clus; 
ANALYSIS: TYPE IS TWOLEVEL; 
MODEL: 
 %WITHIN% 
 fw1 BY y1-y3; 
 fw2 BY y4-y6; 
 fw1 ON x1 x2; 
 fw2 ON x1 x2; 
 %BETWEEN% 
 fb BY y1-y6; 
 fb ON w; 
 
Output: sampstat standardized; 
 


